

Introduction to Type Theory in Agda

Lecture 3 – Dependent types and equality

Todd Waugh Ambridge

10 April 2024

Lecture Outline

1. Recap – Propositional Logic via Types
2. Predicate Logic via Dependent Types
 - ▶ Π and Σ
3. Type universes
 - ▶ Type , Type_1 , Type_2 , \dots
4. Decidable equality
5. Propositional equality
 - ▶ \equiv

Lecture Outline

1. **Recap – Propositional Logic via Types**
2. Predicate Logic via Dependent Types
 - ▶ Π and Σ
3. Type universes
 - ▶ $\text{Type}, \text{Type}_1, \text{Type}_2, \dots$
4. Decidable equality
5. Propositional equality
 - ▶ \equiv

Recap

In the last lecture, we introduced the propositions-as-types interpretation in Agda by showing that function types interpret implication, and further defining:

1. $\mathbb{1}$ for interpreting truth,
2. $\mathbb{0}$ for interpreting falsity,
3. \neg -types for interpreting negation,
4. $+$ -types for interpreting disjunction,
5. \times -types for interpreting conjunction.

Lecture Outline

1. Recap – Propositional Logic via Types
2. **Predicate Logic via Dependent Types**
 - ▶ Π and Σ
3. Type universes
 - ▶ $\text{Type}, \text{Type}_1, \text{Type}_2, \dots$
4. Decidable equality
5. Propositional equality
 - ▶ \equiv

Dependent Types

To complete our propositions-as-types interpretation of constructive logic, we need to interpret the two quantifier connectives of predicate logic:

- ▶ Universal quantification $\forall x : X, P_x$,
- ▶ Existential quantification $\exists x : X, P_x$.

These quantifiers are interpreted by Martin-Lof's dependent types:

- ▶ Π -types interpret “for all” statements,
- ▶ Σ -types interpret “there exists” statements.

MLTT in Agda

- (a) Function types \rightarrow ,
- (b) Natural numbers \mathbb{N} ,
- (c) The unit $\mathbb{1}$ and empty $\mathbb{0}$ types,
- (d) Disjoint union types $+$,
- (e) Binary product types \times ,
- (f) **Dependent function types Π** ,
- (g) Dependent pair types Σ ,
- (h) Type universes $\text{Type}_0, \text{Type}_1, \dots$,
- (i) Identity types \equiv .

Dependent Types – Π -types

Earlier, we introduced the dependent type family, $\text{is-odd} : \mathbb{N} \rightarrow \text{Type}$. But this isn't the first type family we've seen:

- ▶ $- + -, - \times - : \text{Type} \rightarrow \text{Type} \rightarrow \text{Type}$ are binary type families,
- ▶ Each of the induction principles featured functions $P : X \rightarrow \text{Type}$ – these are also type families.

These induction principles also featured dependent functions $p : (x : X) \rightarrow P(x)$.

```
N-induction : -- Type family  
              (P :  $\mathbb{N} \rightarrow \text{Type}$ )  
              → (p0 : P zero)  
              -- Dependent function  
              → (ps : (n :  $\mathbb{N}$ ) → P n → P (succ n))
```

...

Dependent Types – Π -types

Given a type family $P : X \rightarrow \text{Type}$, a *dependent function* $p : (x : X) \rightarrow P(x)$ is a function whose domain type $P(x) : \text{Type}$ *depends* on the value of the given argument $x : X$.

Clearly, as we have already seen a fair few dependent functions, they are built-in to Agda, just like non-dependent functions.

While non-dependent functions $f : A \rightarrow B$ are terms of function types, dependent functions $f : (x : X) \rightarrow Y\ x$ are terms of Π -types.

Dependent Types – Π -types

$$\frac{\Gamma \vdash X : \text{Type} \quad \Gamma, x : X \vdash Y(x) : \text{Type}}{\Gamma \vdash \Pi_{(x:X)} Y : \text{Type}} \quad (\Pi\text{-Form})$$

$$\frac{\Gamma, x : X \vdash y : Y(x)}{\Gamma \vdash \lambda(x : X).y : \Pi_{(x:X)} Y} \quad (\Pi\text{-Intro})$$

$$\frac{\Gamma \vdash f : \Pi_{(x:X)} Y \quad \Gamma \vdash a : X}{\Gamma \vdash f(a) : Y(a)} \quad (\Pi\text{-Elim})$$

$$\frac{\Gamma, x : X \vdash y : Y(x) \quad \Gamma \vdash a : X}{\Gamma \vdash (\lambda(a : A).b)(a) = y[a/x] : B(a)} \quad (\Pi\text{-Comp})$$

Dependent Types – Π -types

We can align Agda's syntax for Π -types with MLTT's.

$\Pi : \{X : \text{Type}\} (Y : X \rightarrow \text{Type}) \rightarrow \text{Type}$

$\Pi \{X\} Y = (\mathbf{x} : X) \rightarrow Y \mathbf{x}$

syntax $\Pi \{X\} (\lambda \mathbf{x} \rightarrow \mathbf{y}) = \Pi \mathbf{x} : X , \mathbf{y}$

Note that non-dependent functions are just special cases of dependent functions, where the type family $P : X \rightarrow \text{Type}$ is constant.

$$(X \rightarrow Y) = \Pi \mathbf{x} : X, Y$$

Dependent Types – Π -types

While non-dependent functions interpret implication, dependent functions interpret universal quantification. That is, to prove $\forall x : X, P x$ holds, we need to define a dependent function $f : \Pi x : X, P x$.

As an example, let's prove that we can decide whether `is-odd n : Type` holds for every $n : \mathbb{N}$. This proof is inductive, following the definition of `is-odd : $\mathbb{N} \rightarrow$ Type` itself.

```
is-decidable : Type  $\rightarrow$  Type  
is-decidable X = X +  $\neg$  X
```

```
odd-or-even :  $\Pi n : \mathbb{N}$  , is-decidable (is-odd n)  
odd-or-even 0 = inr 0-elim  
odd-or-even 1 = inl  $\star$   
odd-or-even (succ (succ n)) = odd-or-even n
```

Dependent Types – Π -types

Let's see another example: first we define the binary type family that corresponds to the order on natural numbers.

```
_<_ : ℕ → ℕ → Type
0     < 0     = 0
0     < succ _ = 1
succ _ < 0     = 0
succ n < succ m = n < m
```

Then, we prove that, for every $n : \mathbb{N}$, it is the case that $n < \text{succ } n$.

```
succ-is-bigger : Π n : ℕ , (n < succ n)
succ-is-bigger zero = ★
succ-is-bigger (succ n) = succ-is-bigger n
```

MLTT in Agda

- (a) Function types \rightarrow ,
- (b) Natural numbers \mathbb{N} ,
- (c) The unit $\mathbb{1}$ and empty $\mathbb{0}$ types,
- (d) Disjoint union types $+$,
- (e) Binary product types \times ,
- (f) Dependent function types Π ,
- (g) **Dependent pair types** Σ ,
- (h) Type universes $\text{Type}_0, \text{Type}_1, \dots$,
- (i) Identity types \equiv .

Dependent Types – Σ -types

Given a type family $Y : X \rightarrow \text{Type}$, how do we interpret the concept that there exists a term $x : X$ such that $Y\ x : \text{Type}$ is true?

The way existential quantification works is the second key difference between constructive and classical logic. Classically, we can show that there exists an $x : X$ that satisfies $Y\ x$ by showing that the lack of such an $x : X$ leads to a contradiction.

But this argument doesn't hold in constructive maths: constructively, to show that $Y\ x$ holds, we have to actually *specify* which $x : X$ is satisfactory.

Dependent Types – Σ -types

So, to show that $\exists x : X, Y x$, we need to provide a pair of terms:

1. A term $x : X$, called the *witness* of Y ,
2. A proof term $Y x : \text{Type}$, which *depends* on the witness.

In MLTT, these dependent pairs are called Σ -types. As with non-dependent pairs (i.e. \times -types), we define them using `record`.

```
record  $\Sigma$  {X : Type} (Y : X  $\rightarrow$  Type) : Type where
  constructor _,_
  field
    fst : X
    snd : Y fst
```

```
syntax  $\Sigma$  {X} ( $\lambda$  x  $\rightarrow$  y) =  $\Sigma$  x : X , y
```


Dependent Types – Σ -types

$$\frac{\Gamma \vdash X : \text{Type} \quad \Gamma, x : X \vdash Y(x) : \text{Type}}{\Gamma \vdash \Sigma_{(x:X)} Y : \text{Type}} \quad (\Sigma\text{-Form})$$

$$\frac{\Gamma, x : X \vdash Y(x) : \text{Type} \quad \Gamma \vdash w : X \quad \Gamma \vdash y : Y(w)}{\Gamma \vdash (w, p) : \Sigma_{(w:X)} Y} \quad (\Sigma\text{-Intro})$$

$$\frac{\Gamma, z : \Sigma_{(x:X)} Y \quad \Gamma, w : X, y : Y(w) \vdash p((w, y)) : P((w, y))}{\Gamma, z : \Sigma_{(x:X)} Y \vdash \Sigma\text{-induction}(P, p, z) : P(z)} \quad (\Sigma\text{-Elim})$$

$$\frac{\Gamma, z : \Sigma_{(x:X)} Y \quad \Gamma, w : X, y : Y(w) \vdash p((w, y)) : P((w, y))}{\Gamma, a : X, b : Y(a) \vdash \Sigma\text{-induction}(P, p, (a, b)) = p((a, b)) : P((a, b))} \quad (\Sigma\text{-Comp})$$

Dependent Types – Σ -types

```
 $\Sigma$ -induction : {X : Type} {Y : X → Type}
  → (P :  $\Sigma$  Y → Type)
  → (p : (w : X) (y : Y w) → P (w , y))
  → (z :  $\Sigma$  Y) → P z
 $\Sigma$ -induction P p (x , y) = p x y
```

We can now re-define \times -types as the non-dependent case of Σ -types (as with non-dependent functions and Π -types).

```
_×_ : Type → Type → Type
A × B =  $\Sigma$  a : A , B
```

```
×-induction =  $\Sigma$ -induction
```

Dependent Types – Σ -types

As an example of using Σ -types, let's prove that, for every $n : \mathbb{N}$, there exists an $m : \mathbb{N}$ larger than it.

```
always-a-bigger-number :  $\prod n : \mathbb{N} , \Sigma m : \mathbb{N} , n < m$   
always-a-bigger-number n = succ n , succ-is-bigger n
```

In the above, we chose to specify $\text{succ } n$ as the witness that there is a number bigger than n . But we could have chose $\text{succ}(\text{succ } n)$ or add 1000 n ...

The term that we choose as a witness changes the computational content of the resulting proof. Therefore, in constructive type theory, the method of proving something is relevant — not just the fact that we have proved it.

This is called *proof relevance*.

Dependent Types – Σ -types

Another important point about Σ -types is that they form collections in our type theory.

For example, the type $\Sigma_{(n:\mathbb{N})} \text{is-odd } n$ collects every possible pair of a number with a proof of its oddness. Therefore, this is the *type of odd numbers* itself.

\mathbb{N}_o : Type

$\mathbb{N}_o = \Sigma \text{ is-odd}$

Dependent Types – Σ -types

But what if we want to collect *types* themselves?

For example, we could carve out a subset of our logic that relates to the Boolean-logic; i.e., we could define a Σ -type that collects all decidable types together.

`DecidableType = Σ X : Type , is-decidable X`

Well, we could, if not for that we get a type error! What is going on here? And how do we fix it? What does `Set1 != Set` mean???

Lecture Outline

1. Recap – Propositional Logic via Types
2. Predicate Logic via Dependent Types
 - ▶ Π and Σ
3. **Type universes**
 - ▶ $\text{Type}, \text{Type}_1, \text{Type}_2, \dots$
4. Decidable equality
5. Propositional equality
 - ▶ \equiv

Type Universes

I've been deliberately vague about what Type itself is. In Martin-Lof's first type theory (which appeared in a 1971 preprint), there were *terms* and there were *types* — terms had types, but types were just types. For example, $a : A$, but $A : \text{Type}$.

This raises the interesting question: what is the type of Type? Well, the 1971 type theory had an axiom that said

$$\text{Type} : \text{Type}.$$

But, similarly to Russell with set theory, Girard showed that this axiom made the system inconsistent.

MLTT in Agda

- (a) Function types \rightarrow ,
- (b) Natural numbers \mathbb{N} ,
- (c) The unit $\mathbb{1}$ and empty $\mathbb{0}$ types,
- (d) Disjoint union types $+$,
- (e) Binary product types \times ,
- (f) Dependent function types Π ,
- (g) Dependent pair types Σ ,
- (h) **Type universes** $\text{Type}_0, \text{Type}_1, \dots$,
- (i) Identity types \equiv .

Type Universes

So Martin-Lof went back to the drawing board, and built his next type theory (1972's MLTT) around the idea of countably-many type universes:

$$\text{Type} : \text{Type}_1 : \text{Type}_2 : \dots$$

A *type universe* is a type whose terms are also types.

Agda also has type universes (but with the annoying name `Set`):

$$\text{Set} : \text{Set}_1 : \text{Set}_2 : \dots$$

Type Universes

So that we don't have to rename a countably infinite number of terms let's properly rename `Set` to `Type` using Agda's builtin file for type universes.

```
open import Agda.Primitive renaming (Set to Type)
```

```
term-of-type-Type1 : Type1
```

```
term-of-type-Type1 = Type
```

```
term-of-type-Type2 : Type2
```

```
term-of-type-Type2 = Type1
```

In that file, we can see a glimpse of how type universes are implemented in Agda. The idea is that `Type`, `Type1`, `Type2`, etc. are actually syntax sugars for the *type universes* `Type lzero`, `Type (lsuc lzero)`, `Type (lsuc (lsuc lzero))`; where these objects beginning with `l` are called *universe levels*.

Type Universes

Now, let's redefine Π and Σ to correctly use type universes.

```
 $\Pi : \{i\ j : \text{Level}\} \{X : \text{Type } i\} (Y : X \rightarrow \text{Type } j)$   
   $\rightarrow \text{Type } (i \sqcup j)$ 
```

```
 $\Pi \{i\} \{j\} \{X\} Y = (x : X) \rightarrow Y\ x$ 
```

```
 $\text{Pi} : \{i\ j : \text{Level}\} (X : \text{Type } i) (Y : X \rightarrow \text{Type } j)$   
   $\rightarrow \text{Type } (i \sqcup j)$ 
```

```
 $\text{Pi } \_ Y = \Pi Y$ 
```

```
syntax  $\text{Pi } X (\lambda x \rightarrow y) = \Pi x : X , y$ 
```

```
infixr -1  $\text{Pi}$ 
```

Type Universes

```
record  $\Sigma$  {i j : Level} {X : Type i} (Y : X  $\rightarrow$  Type j)  
  : Type (i  $\sqcup$  j)
```

```
where
```

```
  constructor _,-
```

```
  field
```

```
    fst : X
```

```
    snd : Y fst
```

```
open  $\Sigma$  public
```

```
Sigma : {i j : Level} (X : Type i) (Y : X  $\rightarrow$  Type j)  
   $\rightarrow$  Type (i  $\sqcup$  j)
```

```
Sigma _ Y =  $\Sigma$  Y
```

```
syntax Sigma X ( $\lambda$  x  $\rightarrow$  y) =  $\Sigma$  x : X , y
```

```
infixr -1 Sigma
```

Type Universes

Now that we have universes, we can define the type of decidable types:

`DecidableType` : Type_1

`DecidableType` = $\Sigma X : \text{Type} , \text{is-decidable } X$

Lecture Outline

1. Recap – Propositional Logic via Types
2. Predicate Logic via Dependent Types
 - ▶ Π and Σ
3. Type universes
 - ▶ $\text{Type}, \text{Type}_1, \text{Type}_2, \dots$
4. **Decidable equality**
5. Propositional equality
 - ▶ \equiv

Decidable equality

We finally have a full, well defined, Type-valued first-order (predicate) logic. The final step is to have an interpretation of equality.

In the second exercise class, we played around with this Type-valued logic. For example, we defined an equality relation on the Booleans...

```
_==_ : Bool → Bool → Type
```

```
tt == tt = 1
```

```
ff == tt = 0
```

```
tt == ff = 0
```

```
ff == ff = 1
```

Decidable equality

...and one on the natural numbers.

```
_≡_ : ℕ → ℕ → Type
zero  ≡ zero  = 1
zero  ≡ succ m = 0
succ n ≡ zero  = 0
succ n ≡ succ m = n ≡ m
```

It is important to realise here that, given any two terms $a, b : \text{Bool}$ (respectively $n, m : \mathbb{N}$), $a == b$ (respectively $n \equiv m$) is a type.

```
term-of-type-Type : Type
term-of-type-Type = 3 ≡ 8
```


Decidable equality

Recall also that, in the exercise class, we showed the equality relation on the natural numbers is indeed an equality relation: it is reflexive, symmetric and transitive.

```
 $\equiv$ -is-reflexive : (n : ℕ) → n ≡ n
```

```
 $\equiv$ -is-reflexive zero = ★
```

```
 $\equiv$ -is-reflexive (succ n) =  $\equiv$ -is-reflexive n
```

```
 $\equiv$ -is-symmetric : (n m : ℕ) → n ≡ m → m ≡ n
```

```
 $\equiv$ -is-symmetric zero zero p = ★
```

```
 $\equiv$ -is-symmetric (succ n) (succ m) p =  $\equiv$ -is-symmetric n m p
```

```
 $\equiv$ -is-transitive : (n m k : ℕ) → n ≡ m → m ≡ k → n ≡ k
```

```
 $\equiv$ -is-transitive zero zero zero p q = ★
```

```
 $\equiv$ -is-transitive (succ n) (succ m) (succ k) p q  
=  $\equiv$ -is-transitive n m k p q
```

Decidable equality

The Booleans and the natural numbers have what we call decidable equality; i.e., given any two terms $a, b : \text{Bool}$ (respectively $n, m : \mathbb{N}$), the question of whether $a == b$ (respectively $n \equiv m$) is a decidable proposition.

This is the same as saying they are decidable types.

```
==-is-decidable : (a b : Bool) → is-decidable (a == b)  
≡-is-decidable : (n m : ℕ) → is-decidable (n ≡ m)
```

Decidable equality

The first proof is by the fact that truth and falsity are decidable propositions; i.e. $\mathbb{1}$ and $\mathbb{0}$ are decidable types.

```
 $\mathbb{1}$ -is-decidable : is-decidable  $\mathbb{1}$ 
```

```
 $\mathbb{1}$ -is-decidable = inl  $\star$ 
```

```
 $\mathbb{0}$ -is-decidable : is-decidable  $\mathbb{0}$ 
```

```
 $\mathbb{0}$ -is-decidable = inr  $\mathbb{0}$ -elim
```

```
==-is-decidable tt tt =  $\mathbb{1}$ -is-decidable
```

```
==-is-decidable tt ff =  $\mathbb{0}$ -is-decidable
```

```
==-is-decidable ff tt =  $\mathbb{0}$ -is-decidable
```

```
==-is-decidable ff ff =  $\mathbb{1}$ -is-decidable
```

Decidable equality

The second proof is also by these facts, and induction.

```
≡-is-decidable zero      zero      = 1-is-decidable
≡-is-decidable zero      (succ m)  = 0-is-decidable
≡-is-decidable (succ n) zero      = 0-is-decidable
≡-is-decidable (succ n) (succ m) = ≡-is-decidable n m
```

Decidable equality

But, as we discussed last lecture, the law of excluded middle¹ does not hold constructively – it is not the case that every proposition is decidable.

It is also not the case that every type has decidable equality; consider the example of functions from last lecture. We cannot decide whether or not two functions $f, g : \mathbb{N} \rightarrow \text{Bool}$ are equal, because any procedure that could do this cannot be guaranteed to halt.

¹This is something that we cannot prove nor disprove in our type theory; we could, if we wanted to (which we don't), add it as an axiom and our theory would remain consistent.

Decidable equality

So, have I lied to you? Because last lecture I entirely motivated the Type-valued logic by saying we could it to define equality on functions; i.e., I said we could define a type family

$$(\mathbb{N} \rightarrow \text{Bool}) \rightarrow (\mathbb{N} \rightarrow \text{Bool}) \rightarrow \text{Type}.$$

But how do we actually go about defining this?

Lecture Outline

1. Recap – Propositional Logic via Types
2. Predicate Logic via Dependent Types
 - ▶ Π and Σ
3. Type universes
 - ▶ $\text{Type}, \text{Type}_1, \text{Type}_2, \dots$
4. Decidable equality
5. **Propositional equality**
 - ▶ \equiv

Propositional equality

We need to think about equality much more generally. In a first-order logic with equality, equality is itself considered to be a proposition.

This suggests that, as with the other connectives of predicate logic, it must be interpreted as a *type (family)*.

Thus far we have interpreted propositional equality differently for each type, but this is not necessary. Rather than thinking about equality as type families

$$Bool \rightarrow Bool \rightarrow \text{Type},$$

or

$$\mathbb{N} \rightarrow \mathbb{N} \rightarrow \text{Type},$$

or

$$(\mathbb{N} \rightarrow Bool) \rightarrow (\mathbb{N} \rightarrow Bool) \rightarrow \text{Type} \dots$$

MLTT in Agda

- (a) Function types \rightarrow ,
- (b) Natural numbers \mathbb{N} ,
- (c) The unit $\mathbb{1}$ and empty $\mathbb{0}$ types,
- (d) Disjoint union types $+$,
- (e) Binary product types \times ,
- (f) Dependent function types Π ,
- (g) Dependent pair types Σ ,
- (h) Type universes $\text{Type}_0, \text{Type}_1, \dots$,
- (i) **Identity types** \equiv .

Propositional equality

...let's just think of it as a single type family.

This type family would have to depend on both the type and the two elements of that type that we are trying to show are equal.

$$\frac{\Gamma \vdash A : \text{Type}_i \quad \Gamma \vdash x : A \quad \Gamma \vdash y : A}{\Gamma \vdash x \equiv_A y : \text{Type}_i} \quad (\equiv\text{-Form})$$

```
data _≡_ {i : Level} {X : Type i} : X → X → Type i
```

Propositional equality

But how would we introduce elements of these types?

When can we genuinely decide that two elements $x, y : X$ of any type $X : \text{Type}$ are equal?

Well... only when they are literally the same thing.

$$\frac{\Gamma \vdash A : \text{Type}_i \quad \Gamma \vdash x : A}{\Gamma \vdash \text{refl } x : x \equiv_A x} \quad (\equiv\text{-Intro})$$

```
data _≡_ {i : Level} {X : Type i} : X → X → Type i where
  refl : (x : X) → x ≡ x
```

Propositional equality

```
data _≡_ {i : Level} {X : Type i} : X → X → Type i where
  refl : (x : X) → x ≡ x
```

By the above, for any two elements $x, y : X$ of the same type $X : \text{Type}$ there is a type $x \equiv y : \text{Type}$ whose terms are identifications of x and y .

These types have one single constructor, which states the reflexivity, which states the reflexivity law of equality: every element $x : X$ is equal to itself.

Therefore, for now, the only way of introducing a term of these types is by writing `refl x : x ≡ x`; but we cannot *prove* that this is the *only* way of identifying two things².

²This is called Axiom K; it is by default on in Agda, but I have switched it off for this course because it is not provable or disprovable in MLTT.

Propositional equality

All we can say for now is that the type $x \equiv y : \text{Type}$ is definitely inhabited if x and y are genuinely (judgementally) equal.

```
five-equals-five : 5  $\equiv$  5
```

```
five-equals-five = refl 5
```

```
adding-1- $\equiv$ -succ : (n :  $\mathbb{N}$ )  $\rightarrow$  add 1 n  $\equiv$  succ n
```

```
adding-1- $\equiv$ -succ n = refl (succ n)
```

Next time...

Next lecture, we will look at the identity type in more detail, exploring how this expands our idea of proof relevance in type theory. Finally, we will see how thinking about equality in MLTT leads us towards univalent type theory.

Please join me in the exercise classes, where you can get experience of programming Type Theory in Agda yourself!